4 - VARIÁVEIS INDEXADAS



Levando-se em conta o seu tipo, as variáveis podem ser d4ididas em INTEIRAS, REAIS E CARÁTER. Pode-se d4idi-las, também levando-se em conta o número de posições de memória a que elas se referem, em VARIÁVEIS SIMPLES e VARIÁVEIS INDEXADAS.

Variáveis Simples: variáveis que criam uma só posição de memória.

EXEMPLO

Todas as que foram até aqui utilizadas.

Variáveis Indexadas: variáveis que criam um conjunto de posições de memória.

As variáveis indexadas são também chamadas de CONJUNTOS, VARIÁVEIS SUBSCRITAS, "ARRAY'S", VETORES (variáveis indexadas com um único índice), MATRIZES (variáveis indexadas com dois índices).

Para ilustrar sua utilização, faz-se a seguir um exemplo que tipifica seu emprego.

4.2 Necessidade da Existência de Variáveis Indexadas

Problema

Os nomes e as alturas de três pessoas são fornecidas em três registros, um por pessoa.

Fazer um algoritmo para imprimir os nomes das pessoas que tenham altura superior à média aritmética das alturas.

A saída e a entrada deste problema estão esquematizadas na Figura 4.1.

Solução 1

Utilizando variáveis simples

Duplicando os registros de dados

SAÍDA ENTRADA

Figura IV-1: Saída e entrada do problema do Item 4-2.



ALGORITMO MEDIAS

C

C CONT, SOMALT: CONTADOR, SOMA DAS ALTURAS

C NOME, ALT: NOME E ALTURA DA PESSOA SENDO PROCESSADA

C MEDIA: MEDIA ARITMETICA DAS ALTURAS

C

INTEIRA CONT, SOMAL

CARATER NOME*20

INICIO

  CONT = 0

  SOMALT = 0

  ENQUANTO (CONT < 3) FACA

    LEIA, NOME, ALT

    SOMALT = SOMALT + ALT

    CONT = CONT + 1

  FIM ENQUANTO

  MEDIA = SOMALT/3

C VERIFICACAO DAS PESSOAS COM ALTURA ACIMA DA MEDIA

C FAZ-SE UMA NOVA LEITURA DOS DADOS (SAO LIDOS DADOS

C  'DUPLICATAS')

  CONT = 0

  ESCREVA, "PESSOAS COM ALTURA SUPERIOR A MEDIA"

  ENQUANTO (CONT < 3) FACA

    LEIA, NOME, ALT

    SE (ALT > MEDIA) ENTAO

      ESCREVA, NOME

    FIM SE

    CONT = CONT + 1

  FIM ENQUANTO

FIM



Observações:

Na primeira estrutura ENQUANTO ... FACA são lidos os dados e acumulada a soma das alturas em SOMAALT. Mas, como se usa uma VARIÁVEL SIMPLES para NOME e outra para ALT, a leitura do 2º registro ocupa na memória o lugar do primeiro, cujos dados são então perdidos. O terceiro apagará os dados do segundo, na memória, e assim sucess4amente, restando no final armazenados apenas os dados do último registro. Para comparar cada altura das pessoas com a MEDIA, uma nova leitura deverá então ser feita, o que acontece na segunda estrutura ENQUANTO ... FACA.

Caso se t4esse 1000 registros de dados, ter-se-ia 1000 duplicatas, o que evidentemente é, no mínimo, desconfortável. O programa acima pouco se alteraria, devendo-se apenas escrever 1000 em lugar de 3 (nas condições dos ENQUANTO ... FACA e no cálculo da MEDIA).



SOLUÇÃO 2

  - Utilizando variáveis simples

  - Armazenando cada dado lido em uma variável.

ALGORITMO MÉDIAS

C

C NOME1, NOME2, NOME3: VARIAVEIS PARA ARMAZENAR NOMES LIDOS

C ALT1, ALT2, ALT3: VARIAVEIS PARA ARMAZENAR ALTURAS LIDAS

C MEDIA: MEDIA ARITMETICA DAS ALTURAS

C

INTEIRA ALT1, ALT2, ALT3

CARATER*20  NOME1, NOME2, NOME3

INICIO

  LEIA, NOME1, ALT1

  LEIA, NOME2, ALT2

  LEIA, NOME3, ALT3

  MEDIA = (ALT1 + ALT2 + ALT3)/3

  ESCREVA, "PESSOAS COM ALTURAS ACIMA DA MEDIA"

  SE (ALT1 > MEDIA) ENTAO

    ESCREVA, NOME1

  FIM SE

  SE (ALT2 > MEDIA) ENTAO

    ESCREVA, NOME2

  FIM SE

  SE (ALT3 > MEDIA) ENTAO

    ESCREVA, NOME3

  FIM SE

FIM

OBSERVAÇÃO

Caso se t4esse 1000 registros de dados, se teriam 1000 leituras e 1000 estruturas SE ... ENTAO, dando um programa de umas 5000 linhas, aproximadamente, contando linhas de comando e linhas de dados. Isto torna esta solução bem mais cara e trabalhosa que a anterior, que também não satisfez.

Parte-se então para uma terceira solução, que utiliza variáveis indexadas.



4.3 Variáveis Indexadas - Vetores (Apêndice A, Item A.7)

SOLUÇÃO 3

Utilizando variáveis indexadas.



INTRODUÇÃO

Suponha-se que, na declaração de tipo, simplesmente  escrevendo-se o nome da variável seguido de um número entre parênteses, seja criado um conjunto de posições de memória. O nome deste conjunto de posições de memória será composto de duas partes: NOME(INDICE). A primeira parte, NOME, indica todo o conjunto. A segunda, que segue a primeira e é posta entre parênteses, é uma variável que, conforme os valores que assume, indica cada uma das posições do conjunto.

EXEMPLO

Declarando

        REAL ALT(30)

      acontece o "milagre": é criado o conjunto de posições posições de memória abaixo:

        ALT(1)  ALT(2) ALT(3) ........ ALT(29) ALT(30)

Diz-se que a variável ALT é indexada ou subscrita e é também         chamada VETOR ou CONJUNTO (Com um índice). O valor do índice indica a que posição do conjunto se está referindo num dado instante.

        ESCREVA, ALT(8), ALT(25)

imprimirá os valores das alturas contidas na 8a. e na 25a. posição de memória do conjunto ALT, pois os índices valem 8 e 25, respect4amente.



ALGORITMO MEDIAS

C

C NOME*20(3): VARIAVEL INDEXADA, COM 3 POSICOES DE MEMORIA,

C          ARMAZENANDO 20 LETRAS POR POSICAO

C ALT(3): VARIAVEL INDEXADA COM 3 POSICOES DE MEMORIA

C CONT: VARIAVEL CONTADORA UTILIZADA PARA CONTROLAR OS 

C "ENQUANTO"

C       E TAMBEM FUNCIONA COMO INDICE

C SOMA, MEDIA: SOMA E MEDIA ARITMETICA DAS ALTURAS, 

C RESPECTIVAMENTE

C

INTEIRA ALT(3), CONT

CARATER NOME*20(3)

INICIO                       					equivale a:

  SOMA = 0                           			!

  CONT = 1                           			!  LEIA, NOME(1), ALT(1)

  ENQUANTO (CONT  <= 3) FACA         	!  SOMA = SOMA + ALT(1)

    LEIA, NOME(CONT), ALT(CONT)      	!  LEIA, NOME(2), ALT(2)

    SOMA = SOMA + ALT(CONT)          		!  SOMA = SOMA + ALT(2)

    CONT = CONT + 1                  		!  LEIA, NOME(3), ALT(3)

  FIM ENQUANTO                       		!  SOMA = SOMA + ALT(3)

  MEDIA = SOMA/3                     			!

  CONT = 1                           			! equ4ale a:

C PESSOAS COM ALTURA ACIMA DA MEDIA 

  ENQUANTO (CONT <= 3) FACA          	!   SE (ALT(1) > MEDIA) ENTAO

    SE (ALT(CONT) > MEDIA) ENTAO 		!     ESCREVA, NOME(1)

      ESCREVA, NOME(CONT)            		!   FIM SE

    FIM SE                           			!   SE (ALT(2) > MEDIA) ENTAO

    CONT = CONT + 1                  		!     ESCREVA, NOME(2)

  FIM ENQUANTO                       		!   FIM SE

FIM                                  				!   SE (ALT(3) > MEDIA) ENTAO

                                     				!     ESCREVA, NOME(3)

                                     				!   FIM SE



Observações:

Caso fossem 1000 registros, em vez de 3, bastaria substituir  3 por 1000.

Como se criam desta maneira, facilmente, muitas posições de memória, é fácil ultrapassar a capacidade da mesma, sobretudo quando se usam variáveis com mais de 1 índice (ver Item 4.5 - MATRIZES)

A variável CONT serve de índice das variáveis ALT e NOME e como tal, deve ser do tipo INTEIRO, além de só poder assumir valores positivos pois os índices de ALT e de NOME vão de 1 a 3 (neste algoritmo).



EXEMPLO 2



  PROBLEMA

Ler um conjunto de 10 números inteiros, dados um em cada registro, e imprimi-los em ordem inversa à lida.



SOLUÇÃO

Este é outro exemplo típico de aplicação de variáveis indexadas, pois os números dados são de mesma natureza e portanto podem ser armazenados em um conjunto (variável indexada) e, mais que isso, devem ser armazenados obrigatoriamente, pois só após lido o último registro se poderá iniciar a impressão.



ALGORITMO INVERTE

C

C PROGRAMADOR: ZE FERINO - 22/09/89

C INDICE: VARIAVEL QUE CONTROLA O INDICE DA VARIAVEL NUMERO

C

INTEIRA NUMERO(10), INDICE

INICIO

  PARA INDICE DE 1 ATE 10 FACA

    LEIA, NUMERO(INDICE)

  FIM PARA

C IMPRESSAO EM ORDEM INVERSA A LIDA

  PARA INDICE DE 1 ATE 10 FACA

    ESCREVA, NUMERO(11-INDICE)

  FIM PARA

FIM



4.4 Estrutura de Repetição Para...Faça Embutida

(Apêndice A, Item 7.1)

INTRODUÇÃO

    No exemplo anterior, verifica-se que existe um grande desperdício de espaço em cada registro quando nele se escreve apenas um número, ou poucos dados, de modo geral. Mais racional seria fornecer os 10 números em um só registro. Suponha-se então que um só registro contenha os 10 números e se tenha que lê-los e armazená-los.

4.4.1 Leitura com Estrutura de Repetição Para ... Faça Embutida (Apêndice A, Item 7)

Como se tem um só registro, apenas um comando LEIA poderá ser dado e este único comando deverá armazenar os 10 números.

  Uma solução seria comandar:

LEIA, NUMERO(1), NUMERO(2), NUMERO(3), ..., NUMERO(10)



Pode-se ver que o nome NUMERO do conjunto vetorial de 10 posições é constante, apenas seu índice varia de 1 a 10, o que leva à idéia de se utilizar como índice uma variável, e dizer no próprio comando que esta variável vai de 1 a 10.

Assim,

         LEIA, NUMERO(INDICE) PARA INDICE DE 1 ATE 10



FORMA GERAL DA ESTRUTURA DE REPETIÇÃO

PARA ... FACA EMBUTIDA NA LEITURA:

LEIA, variável(índice) PARA índice DE início ATE fim

onde:

Variável: nome da variável indexada (que deve ter sido previamente             declarada com p posições)

Índice: variável inteira que assume valores entre 1 e p (p: número de posições reservado para "variável" na declaração de tipo)

Ínício: variável ou constante inteira que indica o primeiro valor de "índice"

Fim: variável ou constante inteira que indica o último valor de "índice"




4.4.2
 
I
mpressão com 
E
strutura de 
R
epetição 
P
ara ... 
F
aça 
E
mbutida



O mesmo desperdício de espaço em que se incorria na leitura, fornecendo apenas um número por registro, repete-se na impressão dos 10 números, pondo um por linha, o que deixará talvez mais de 90% de cada linha sem utilização.



Novamente o racional seria imprimir os 10 números na mesma linha, o que só poderia ser conseguido com um único comando de impressão. Uma solução "lusitana" para isto seria:



     ESCREVA, NUMERO(1), NUMERO(2), NUMERO(3), ..., NUMERO(10)


Aproveitando-se a mesma idéia do PARA ... FAÇA embutido na leitura para a impressão, i. é, dando ao índice o nome de uma variável e dizendo que esta variável vai de 1 a 10 se tem:



    ESCREVA, NUMERO(INDICE) PARA INDICE DE 1 ATE 10


Forma G
eral da 
E
strutura de 
R
epetição 
P
ara ... Faca 
E
mbutida na 
I
mpressão:


     ESCREVA, variável(índice) PARA índice DE início ATE fim

onde,


variável, índice, início e fim têm o mesmo significado que o PARA ... FACA embutido de leitura.



Observação
:



  
A estrutura PARA ... FACA embutida só pode ser utilizada para leitura e impressão (Comandos LEIA e ESCREVA). Então quando se quisesse, por exemplo, somar os 10 números lidos anteriormente, o trecho de algoritmo abaixo:



  SOMA = 0

  SOMA = SOMA + NUMERO(INDICE) PARA INDICE DE 1 ATE 10

não é permitido, pois é um comando de atribuição. Neste caso o correto seria usar a estrutura PARA ... FACA normal (explícita, não embutida):

  SOMA = 0

  PARA INDICE DE 1 ATE 10 FACA

    SOMA = SOMA + NUMERO(INDICE)

  FIM PARA


O algoritmo INVERTE do item 4.3.- usando PARA ... FACA embutido na leitura e na impressão, ficaria:



ALGORITMO INVERTE

C PROGRAMADOR: ZE FERINO - 29/09/89 

C UTILIZACAO DE PARA ... FACA EMBUTIDO (LEITURA E IMPRESSAO)

C INDICE: VARIAVEL QUE CONTROLA O INDICE DA VARIAVEL NUMERO

INTEIRA NUMERO(10)

INICIO

  LEIA, NUMERO(INDICE) PARA INDICE DE 1 ATE 10

C IMPRESSAO EM ORDEM INVERSA

  ESCREVA, NUMERO(11-INDICE) PARA INDICE DE 1 ATE 10

FIM

4.5
 
Variáveis Bi-Indexadas o
u Matrizes



Um caso mais geral das variáveis indexadas vistas no item anterior, que tinham apenas um índice, e por isto são conhecidas como VETORES, é o das variáveis com dois índices, conhecidas como MATRIZES. As MATRIZES aplicam-se sobretudo nos casos em que os dados podem ser arranjados em tabelas, onde cada dado é ind4idualizado por um primeiro índice que localiza a LINHA e por um segundo que localiza a COLUNA da tabela em que se encontra. Valem para estes índices as mesmas regras que para os VETORES.



PROBLEMA


Um fazendeiro anotou as produções de leite, em litros, de cada uma de suas dez melhores vacas, durante 30 dias e montou a Tabela 4-1 abaixo esquematizada:





VACA

DIA�1�2�3�4�5�6�7�8�9�10��1�15.5�18.5�10.2�18.3�20.3�15.2�9.5�13.2�11.2�23.4��2�18.2�15.2�9.9�15.0�19.9�20.2�12.9�15.4�13.4�25.4��3�16.2�12.2�10.9�16.2�20.0�17.2�13.3�16.2�12.2�24.9��������������30�15.6�11.0�10.9�16.3�18.9�17.2�15.2�14.2�12.6�23.9��Tabela 4-1: Produções de leite

  Fazer um algoritmo que:

    a) Leia e armazene os dados

    b) Imprima as produções da vaca 5 e a da vaca 8 no mês

    c) Imprima a produção total do dia 10 com a do dia 20

    d) Quantos litros de leite o fazendeiro tirou no mês.

SOLUÇÃO


As produções de leite acima são, claramente, dados de mesma natureza e podem ser armazenados em uma matriz tipo real de 30 linhas por 10 colunas, perfazendo um total de 300 posições de memória. Seja LEITE o nome da matriz, que então será declarada como: REAL LEITE (30,10).



a) Leitura e armazenamento dos dados.

   LEIA, LEITE(1,1), LEITE(1,2), LEITE(1,3), ..., LEITE(1,10)

   LEIA, LEITE(2,1), LEITE(2,2), LEITE(2,3), ..., LEITE(2,10)

    .

   LEIA, LEITE(30,1), LEITE(30,2), LEITE(30,3), ..., LEITE(30,10)

   	Usando uma estrutura de repetição, e chamando de LINHA o primeiro índice, que varia de 1 a 30, os 30 comandos LEIA acima ficam:

   PARA LINHA DE 1 A 30 FACA

     LEIA, LEITE(LINHA,1), LEITE(LINHA,2), ......., LEITE(LINHA,10)

   FIM PARA


Vê-se que o segundo índice, que indica a coluna em que se encontra o elemento na tabela, varia de 1 a 10, para cada valor assumido por LINHA, isto é:



  se LINHA = 1, o segundo índice assume valores 1, 2, 3, ..., 10

  se LINHA = 2, o segundo índice assume valores 1, 2, 3, ..., 10

  etc., até LINHA = 30.


Para generalizar o processo de variação de 1 a 10 das colunas, dá-se a elas o nome de uma variável, COLUNA, por exemplo, e se diz que esta variável vai de 1 a 10, dentro do próprio comando LEIA, isto é, usa-se uma estrutura PARA ... FACA embutida.



Assim a leitura da matriz acima ficaria:



  PARA LINHA DE 1 ATE 30 FACA

    LEIA LEITE(LINHA, COLUNA) PARA COLUNA DE 1 A 10

  FIM PARA


Observação
:



Como se observa, cada execução do comando LEIA lê uma linha da matriz.


Para que a leitura acima funcione, supõe-se que cada linha da matriz está escrita em um registro.


Sugere-se que o programador iniciante leia suas matrizes sempre desta forma, que é a adotada neste trabalho.




b) Imprima as produções da vaca 5 e a da vaca 8 no mês


O que se deseja na realidade é a soma da 5a. coluna e a soma da 8a. coluna. Então basta zerar um acumulador para a vaca 5 e outro para a vaca 8 e percorrer todas as linhas da matriz somando os elementos da coluna 5 e os da coluna 8, isto é, o primeiro índice, que indica a linha, varia de 1 a 30, equanto o segundo fica constante valendo ou 5 ou 8.



  VACA5=0

  VACA8=0

  PARA LINHA DE 1 ATE 30 FACA

    VACA5 = VACA5 + LEITE(LINHA,5)

    VACA8 = VACA8 + LEITE(LINHA,8)

  FIM PARA



c) Imprima a produção total do dia 10 com a do dia 20

SOLUÇÃO

	Neste caso deve-se somar as produções do dia 10 com as do dia 20, das 10 vacas, isto é, percorrer as linhas 10 e 20 somando as colunas de 1 a 10 destas linhas, e acumulando os valores.

  DIA10E20 = 0

  PARA COLUNA DE 1 ATE 10 FACA

    DIA10E20 = DIA10E20 + LEITE(10, COL) + LEITE(20, COL)

  FIM PARA



d) Quantos litros de leite o fazendeiro tirou no mês.

	A produção total de leite do mês é a soma de todos os elementos da matriz, isto é, deve-se percorrer cada linha e em cada linha todas as suas colunas somando os elementos lá posicionados em um acumulador. Então a variável LINHA assumirá valores de 1 a 30 e para cada valor assumido por LINHA, a variável COLUNA assumirá valores de 1 a 10.

  DIA01A30 = 0

  PARA LINHA DE 1 A 30 FACA

    PARA COLUNA DE 1 A 10 FACA

      DIA01A30 = DIA01A30 + LEITE(LINHA, COLUNA)

    FIM PARA

  FIM PARA



SOLUÇÃO COMPLETA:

ALGORITMO LEITE

C LEITE(30,10): MATRIZ DAS PRODUCOES DE LEITE DE 10 VACAS, POR

C MES

C VACA5, VACA8: PRODUCAO DA VACA N. 5 E DA N. 8

C DIA10E20, DIA01A30: PRODUCOES DE LEITE DOS DIAS 10 E 20; E DO

C MES

REAL LEITE(30,10), VACA5, VACA8, DIA10E20, DIA01A30

INTEIRA LINHA, COLUNA

INICIO

C INICIALIZACAO DOS ACUMULADORES

  VACA5 = 0

  VACA8 = 0

  DIA10E20 = 0

  DIA01A30 = 0

C LEITURA E ARMAZENAMENTO DA MATRIZ

  PARA LINHA DE 1 A 30 FACA

    LEIA, LEITE(LINHA, COLUNA) PARA COLUNA DE 1 A 30

  FIM PARA

C PRODUCOES DA VACA 5 E DA VACA 8

  PARA LINHA DE 1 ATE 30 FACA

    VACA5 = VACA5 + LEITE(LINHA,5)

    VACA8 = VACA8 + LEITE(LINHA,8)

  FIM PARA

C PRODUCAO TOTAL DO DIA 10 COM A DO DIA 20

  PARA COLUNA DE 1 ATE 10 FACA

    DIA10E20 = DIA10E20 + LEITE(10, COL) + LEITE(20, COL)

  FIM PARA

C PRODUCAO TOTAL DO MES

  PARA LINHA DE 1 A 30 FACA

    PARA COLUNA DE 1 A 10 FACA

      DIA01A30 = DIA01A30 + LEITE(LINHA, COLUNA)

    FIM PARA

  FIM PARA

C IMPRESSAO DOS RESULTADOS

  ESCREVA, "PRODUCAO DA VACA NUMERO 5 ............:", VACA5

  ESCREVA, "PRODUCAO DA VACA NUMERO 8 ............:", VACA8

  ESCREVA, "PRODUCAO DO DIA 10 COM A DO DIA 20 ...:", DIA10E20

  ESCREVA, "PRODUCAO TOTAL DO MES ................:", DIA01A30

FIM

4.6
 
Exercícios Resolvidos


4.6.1
 
Exercícios Resolvidos - Vetores



Exemplo
 1 - 
Ordenação



Introdução



Conforme discussão iniciada no item III-6 a ordenação é um dos processamentos mais executados pelos computadores, portanto é necessário que se encontrem métodos eficientes para que não se perca muito tempo, sobretudo quando a ordenação é feita sobre grandes conjuntos de dados, como uma lista telefônica, por exemplo.



Existem d4ersas metodologias para ordenar dados e uma delas foi sugerida no item 3-6. Ela não foi lá programada porque o ordenamento de um conjunto exige que todos os seus elementos estejam na memória, e isto é mais facilmente conseguido com o uso de variáveis indexadas, assunto visto neste capítulo.



Um segundo método, chamado método da bolha, compara os dois primeiros elementos do conjunto e troca sua posição caso não estejam ordenados. A seguir compara o segundo com o terceiro elementos e novamente troca-os, se for o caso. Assim sucess4amente.



Por exemplo:

          21        12        7         5

���          						(primeira comparação)



          12        21         7         5 		(novo conjunto)

���                   					(segunda comparação)



��          12        7         21         5 		(novo conjunto)

�                             				(terceira comparaç ão)



          12        7         5         21  		(novo conjunto)




Note-se que após 3 (n-1 se o conjunto t4er n elementos) comparações pelo menos um elemento estará em seu lugar (neste caso o maior). Repetindo-se o processo uma segunda vez, arrastar-se-á o número 12 para sua posição e uma terceira ordenará os números 5 e 7.



Assim para ordenar n números podem ser necessárias n-1 repetições do processo de comparar 2 a 2 os elementos, até o final (Como em cada vez se fazem n-1 comparações, podem ser feitas (n-1)*(n-1) = n**2 - 2n - 1 comparações). Este número de comparações só será necessário na pior das hipóteses, como no caso do conjunto acima, que estava ordenado na ordem exatamente oposta à pretendida. Se o conjunto original fosse: 5  7  21  12 uma única passagem ordenaria o conjunto. Assim, se na i-ésima passagem, 1=<i<=n-1 não ocorrer nenhuma troca entre dois números consecut4os, conclui-se que o conjunto já está ordenado e não é mais necessário repetir o processo (economizando-se n-1-i passagens).



Outra melhoria no processo consiste em economizar uma comparação em cada passagem pelo conjunto. De fato, na primeira passagem o último elemento fica em sua posição correta. Assim, na segunda passagem não é necessário comparar o penúltimo com o último. Na terceira passagem pode-se economizar duas comparações (já que o último e o penúltimo estarão ordenados). E assim sucess4amente.



Estes dois melhoramentos não estaõ introduzidos no algoritmo exemplo feito abaixo.





Exemplo: Para cada aluno matriculado em Computação faz-se um                  registro com seu nome e seu número de matrícula.

         Fazer um algoritmo para imprimir uma relação por orem alfa-bética, conforme esquema abaixo:



         ORDEM    	 NOME                   		 MATRICULA

          1.-      XXXXXXXXXXXXXXXXXXXXX     999999999

          2.-      XXXXXXXXXXXXXXXXXXXXX     999999999

           .

           .

  	99.-    XXXXXXXXXXXXXXXXXXXXX     999999999



ALGORTIMO LISTAORDEM

INTEIRA I, TOTAL, VEZ

CARATER NOME*35(400), MATRICULA*8(400), AUX*35

INICIO

C LEITURA E ARMAZENAMENTO DOS DADOS

  I = 1

  LEIA, NOME(I), MATRIC(I)

  ENQUANTO (NOME(I) # "XYZ") FAÇA

    I = I+1

    LEIA, NOME(I), MATRIC(I)

  FIM ENQUANTO

  TOTAL = I-1

C REPETICAO DO PROCESSO TOTAL-1 VEZES

  PARA VEZ DE 1 ATE TOTAL-1 FAÇA

    PARA I DE 1 ATE TOTAL-1 FAÇA

C     CASO HAJA DOIS NOMES CONSECUT4OS EM 'DESORDEM' TROCA-LOS

C     AS MATRICULAS TAMBEM DEVEM SER TROCADAS

      SE ( NOME(I) > NOME(I+1) ) ENTAO

        AUX = NOME(I)

        NOME(I) = NOME(I+1)

        NOME(I+1) = AUX

        AUX = MATRIC(I)

        MATRIC(I) = MATRIC(I+1)

        MATRIC(I+1) = AUX

      FIM SE

    FIM PARA

  FIM PARA

C IMPRESSAO DA LISTA ORDENADA

  ESCREVA, "ORDEM       NOME           MATRICULA

  PARA I DE 1 ATE TOTAL FAÇA

    ESCREVA, I, NOME(I), MATRIC(I)

  FIM PARA

FIM



4.6.2
 
E
xercícios 
R
esolvidos - 
V
etores 


ALGORITMO EX.4-7-2

C PROGRAMADOR ZE; 26/09/89

INTEIRA NUM(100),I,REGISTRO

C REGISTRO: CONTROLA O REGISTRO QUE ESTA SENDO LIDO

C REGIMAX, REGIMIN: REGISTROS DO MAIOR E MENOR NUMERO

C NUM(100): VARIAVEL QUE ARMAZENA OS 100 NUMEROS LIDOS

C I: VARIAVEL INDICE; VAI DE 1 A 100

C I: VARIAVEL DE CONTROLE DO REGISTRO LIDO (VAI DE 1 A 10)

INICIO

C LEITURA E ARMAZENAMENTO DOS DADOS

  PARA REGISTRO DE 1 ATE 10 FACA

    FIM   =REGISTRO*10

    INICIO=REGISTRO*10-9

    LEIA, NUM(I) PARA I DE INICIO ATE FIM

  FIM PARA

C ENCONTRAR MAIOR (MAX), MENOR(MIN), BEM COMO SUAS POSICOES (IND...)

  MAX=NUM(1)

  INDMAX=1

  MIN=NUM(1)

  INDMIN=1

  PARA I DE 2 ATE 100 FACA

    SE (NUM(I)>MAX) ENTAO

      MAX=NUM(I)

      INDMAX=I

    FIM SE

    SE (NUM(I) < MIN) ENTAO

      MIN=NUM(I)

      INDMIN=I

    FIM SE

  FIM PARA

  REGISMAX=(INDMAX-1)/10+1

  REGISMIN=(INDMIN-1)/10+1

 ESCREVA, 'MAIOR NUMERO, INDICE E REGISTRO',MAX,INDMAX,REGISMAX

  ESCREVA, 'MENOR NUMERO, INDICE E REGISTRO',MIN,INDMIN,REGISMIN

FIM



ALGORITMO EX 4-7-5 - PRAIAS

C PROGRAMADOR ZE; 26/09/89

INTEIRA NUMTUR(10)/10*0/

REAL RMED(10)/10*0.0/,SALTUR(10)/10*0.0/

C NUMTUR(10): NUM. TURISTAS CADA UMA DAS 10 PRAIAS

C RMED(10): RENDA MEDIA DOS TURISTAS DE CADA UMA DAS 10 PRAIAS

C SALTUR(10): TOTAL DOS SALARIOS DOS TUR. DE CADA UMA DAS 10

C PR.

INICIO

C LEITURA DOS DADOS, CONTAGEM DOS TUR. POR PRAIA E 

CTOTALIZAC. DAS RENDAS POR PRAIA

  LEIA, NUMPRAIA,RENDA

  ENQUANTO (NUMPRAIA NAO IGUAL 11) FACA

    NUMTUR(NUMPRAIA)=NUMTUR(NUMPRAIA)+1

    SALTUR(NUMPRAIA)=SALTUR(NUMPRAIA)+RENDA

    LEIA, NUMPRAIA,RENDA

  FIM ENQUANTO

C CALCULO E IMPRESSAO DAS RENDAS DAS 10 PRAIAS

  PARA NUMPRAIA DE 1 ATE 10 FACA

    SE (NUMTUR(NUMPRAIA) = 0) ENTAO

      ESCREVA, 'NAO HOUVE TURISTAS NA PRAIA',NUMPRAIA

    SENAO

      RMED(NUMPRAIA)=SALTUR(NUMPRAIA)/NUMTUR(NUMPRAIA)

      ESCREVA, 'RENDA MEDIA DOS TURISTAS DA PRAIA ===>',NUMPRAIA

      ESCREVA, '===>',NUMTUR(NUMPRAIA),RMED(NUMPRAIA)

    FIM SE

  FIM PARA

FIM



ALGORITMO EX.4-7-8

C PROGRAMADOR ZE; 26/09/89

INTEIRA I,N

REAL FX(100),INTEGRAL,H

C I: VARIAVEL QUE INDICA O I-ESIMO PONTO. N: INDICA O NUM. DE

C PTOS.

C FX(100): VALOR DA FUNCAO F(X) EM CADA PONTO. ARBITROU-SE NO 

C MAX.100 PONTOS. N INDICA O NUM. EXATO DE PTOS. CONSIDERADOS

C INTEGRAL: INDICA O VALOR CALCULADO DA INTEGRAL

C H: DISTANCIA ENTRE DOIS PONTOS CONSECUTIVOS DE X

C

INICIO

C LEITURA DOS DADOS

  LEIA, N

  LEIA, H,FX(I), (I=1,(N+1))

  PARA I DE 2 ATE N FACA

    INTEGRAL = INTEGRAL + 2*FX(I)

  FIM PARA

  INTEGRAL = (INTEGRAL + FX(1) + FX(N+1)) * H/2

  ESCREVA, 'O VALOR DA INTEGRAL DE F(X) EH =', INTEGRAL

FIM

 4.6.2
 
Exercícios Resolvidos - Matrizes



Exemplo
 1 - 
Geração 
d
e Matrizes



Fazer um algoritmo que gere e imprima uma matriz 5x5 tal que os elementos da linha i e coluna j valham o dobro da linha mais o triplo da coluna, exceto os da 5a. linha, os da 5a. coluna e os da diagonal principal, que devem valer 0.



Solução



Como não se fornecem dados, todos os elementos devem ser gerados na própria memória conforme a regra fornecida, preenchendo toda a matriz. Em seguida pode-se ir na linha 5, na coluna 5 e na diagonal principal e zerar as posições.



ALGORITMO DOIS

C PROGRAMADOR ZE; 26/09/89

INTEIRA MATRIZ(5,5), LINHA, COLUNA

INICIO

C GERACAO DA MATRIZ "COMPLETA"

  PARA LINHA DE 1 ATE 5 FACA

    PARA COLUNA DE 1 A 5 FACA

      MATRIZ(LINHA, COLUNA) = 2*LINHA + 3*COLUNA

    FIM PARA

  FIM PARA

  PARA INDICE DE 1 ATE 5 FACA

C   ZERAR LINHA 5

    MATRIZ(5, INDICE) = 0

C   ZERAR COLUNA 5

    MATRIZ(INDICE, 5) = 0

C   ZERAR DIAGONAL PRINCIPAL

    MATRIZ(INDICE, INDICE) = 0

  FIM PARA

C IMPRESSAO DA MATRIZ

  PARA LINHA DE 1 ATE 5 FACA

    ESCREVA, MATRIZ(LINHA, COLUNA) PARA COLUNA DE 1 ATE 5

  FIM PARA

FIM



EXEMPLO 2 - MULTIPLICAÇÃO DE MATRIZES

	Cada elemento P(i,j) da matriz P(m,l), produto das matrizes A(m,n).B(n,l), é definido por:

     P(i,j) = A(i,k).B(k,j),  para: i = 1, ..., m

                                    j = 1, ..., l

                                    k = 1, ..., n

	Supondo-se que as matrizes A e B tenham dimensões máximas de 100x100, tem-se o seguinte algoritmo:



ALGORITMO MULMATRIZ

INTEIRA I,J,K,M,N,L

REAL A(100,100), B(100,100), P(100,100)

INICIO

  PARA I DE 1 ATE M FACA

    PARA J DE 1 ATE L FACA

      P(I,J) = 0.

      PARA K DE 1 ATE N FACA

        P(I,J) = P(I,J) + A(I,K) . B(K,J)

      FIM PARA

    FIM PARA

  FIM PARA

C IMPRESSAO DAS TRES MATRIZES

  PARA I DE 1 ATE M FACA

    ESCREVA, A(I,J) PARA J DE 1 ATE N

  FIM PARA

  PARA I DE 1 ATE N FACA

    ESCREVA, B(I,J) PARA J DE 1 ATE L

  FIM PARA

  PARA I DE 1 ATE M FACA

    ESCREVA, P(I,J) PARA J DE 1 ATE L

  FIM PARA

FIM



EXEMPLO 3

ALGORITMO EX.4.7.2- 5

C PROGRAMADOR ZE; 26/09/89

C

INTEIRA QUEST,ALUNO

REAL GAB(15),RESP(30,15),NOTA(30)

CARATER NOME(30)*25

INICIO

C LEITURA DO GABARITO E DAS RESPOSTAS

  LEIA GAB(QUEST) PARA QUEST DE 1 ATE 15

  PARA ALUNO DE 1 ATE 30 FACA

    LEIA, NOME(ALUNO), RESP(ALUNO,QUEST) PARA QUEST DE 1 ATE 15

  FIM PARA

C CALCULO DAS NOTAS E IMPRESSAO

  PARA ALUNO DE 1 ATE 30 FACA

    NOTA(ALUNO)=0

    PARA QUEST DE 1 ATE 15 FACA

      SE (RESP(ALUNO,QUEST) = GAB(QUEST)) ENTAO

        NOTA(ALUNO) = NOTA(ALUNO) + 10./15

      FIM SE

    FIM PARA

    ESCREVA, ALUNO,NOME(ALUNO),NOTA(ALUNO)

  FIM PARA

C ORDENACAO DECRESCENTE DAS NOTAS

  .

C IMPRESSAO DA LISTA

  .

FIM

    

EXEMPLO 4 - EX. Nº 7



ALGORITMO EX.4.7.2- 7

C PROGRAMADOR ZE; 26/09/89

C

INTEIRA LIN,COL,NLIN,NCOL,

REAL A(100,100),SOMADIAG(100)

INICIO

C LEITURA DA ORDEM DA MATRIZ E DE SEUS ELEMENTOS

  LEIA, NLIN,NCOL

  PARA LIN DE 1 ATE NLIN FACA

    LEIA, A(LIN,COL) PARA COL DE 1 ATE NCOL

  FIM PARA

C CALCULO DAS SOMAS DOS 'ELEMENTOS NAO-DIAGONAIS'

  PARA LIN DE 1 ATE NLIN FACA

    SOMADIAG(LIN) = 0

    PARA COL DE 1 ATE NCOL FACA

C     CONSIDERAR MODULOS

      SE (A(LIN,COL) >= 0) ENTAO

        SOMADIAG(LIN) = SOMADIAG(LIN) + A(LIN,COL)

      SENAO

        SOMADIAG(LIN) = SOMADIAG(LIN) + A(LIN,COL) * (-1)

      FIM SE

    FIM PARA

C   SUBTRACAO DO 'ELEMENTO DIAGONAL'

    SOMADIAG(LIN) = SOMADIAG(LIN) - A(LIN,LIN)

  FIM PARA

C VERIFICAR EM QUANTAS LINHAS A SOMA EH MENOR E EM QTAS. EH IGUAL

  NMENOR = 0; NIGUAL = 0

  PARA LIN DE 1 ATE NLIN FACA

    SE ( SOMADIAG(LIN) < A(LIN,LIN) ) ENTAO

      NMENOR = NMENOR + 1

    SENAO

      SE ( SOMADIAG(LIN) = A(LIN,LIN) ) ENTAO

        NIGUAL = NIGUAL + 1

      FIM SE

    FIM SE

  FIM PARA

  SE ( NMENOR + NIGUAL = NLIN  E  NIGUAL > 0 )ENTAO

    ESCREVA, 'A MATRIZ A EH DIAGONAL DOMINANTE'

  SENAO

    ESCREVA, 'A MATRIZ A NAO E DIAGONAL DOMINANTE'

  FIM SE

FIM 



ALGORITMO EX.4.7.2 - 14

C PROGRAMADOR ZE; 26/09/89

C

INTEIRA LIN,COL,NLIN,NCOL,PESO,SOMAPESO

REAL NOTA(20000,7),SOMA(7),MEDIAG(7)

CARATER NOME*25(30)

C

INICIO

C LEITURA E ARMAZENAMENTO DOS DADOS

  LIN=1

  LEIA, NOME(LIN),NOTA(LIN,COL) PARA COL DE 1 ATE 7

  ENQUANTO (NOME(LIN) NAO IGUAL 'XYZ') FACA

    LIN= LIN+1

    LEIA, NOME(LIN),NOTA(LIN,COL) PARA COL DE 1 ATE 7

  FIM ENQUANTO

C 'DESCONTO' DO REGISTRO BANDEIRA E CALCULO DAS MEDIAS GERAIS

  NLIN=NLIN-1

  PARA COL DE 1 ATE 7 FACA

    PARA LIN DE 1 ATE NLIN FACA

      SOMA(COL)=SOMA(COL)+NOTA(LIN,COL)

    FIM PARA

    MEDIAG((COL)=SOMA(COL)/NLIN

  FIM PARA

C CALCULO DA NOVA MATRIZ DE NOTAS, COM OS PESOS

  SOMAPESO=0

  PARA COL DE 1 ATE 7 FACA

    SE (MEDIAG(COL)<5) ENTAO

      PESO=3

    SENAO

      SE (MEDIAG(COL)<7) ENTAO

        PESO=4

      SENAO

        PESO=5

      FIM SE

    FIM SE

    PARA LIN DE 1 ATE NLIN FACA

      NOTA(LIN,COL)=NOTA(LIN,COL)*PESO

    FIM PARA

    SOMAPESO=SOMAPESO+PESO

  FIM PARA

C CALCULO DAS NOTAS DE CADA ALUNO

  PARA LIN DE 1 ATE NLIN FACA

    SOMA=0

    PARA COL DE 1 ATE 7 FACA

      SOMA=SOMA+NOTA(LIN,COL)

    FIM PARA

    MEDIA(LIN)=SOMA/SOMAPESO

  FIM PARA

C ORDENACAO DAS NOTAS DOS ALUNOS

  .

  .  (Ver exemplo LISTAORDEM - ITEM IV.6.1)

  .

C IMPRESSAO DOS RESULTADOS

  

4.7
 
Exercícios Propostos


4.7.1
 
Vetores



Fazer um algoritmo que leia um conjunto de 10 números em um registro e imprima-os em ordem inversa à lida.


Resolva o problema da inversaõ usando ENQUANTO ... FAÇA e PARA ... FAÇA.


Suponha que se tem 100 números, 10 em cada registro. Faça a leitura de modo que o conjunto dos 100 números fique armazenado em um vetor de 100 posições. O mesmo programa deve calcular a diferença entre o maior e o menor número, bem como as posições (índices de 1 a 100) que os mesmos ocupam no conjunto, e os registros (número de ordem) em que foram lidos.


Para um aluno ser considerado aprovado em uma determinada disciplina é preciso que tenha nota igual ou superior à nota média da turma. Sabendo que a turma é composta de 10 alunos e que foram dados o nome e a nota de cada um, faça um algoritmo que imprima o nome dos alunos aprovados.


Fazer um algoritmo que leia o nome, o telefone e o sexo ('M' = Masculino, 'F' = Feminino) de um conjunto de pessoas, sócias de um clube e imprima:


a) Lista de nomes e telefones, ordenada alfabeticamente por nomes.


b) Lista de nomes e telefones, ordenada por telefones, em ordem crescente.

c) Crie o vetor MASCULINO contendo os nomes dos homens e o vetor FEMININO, contendo o nome das mulheres, e depois, ordene-os alfabeticamente.



Oberva
ç
ão
.: Use no seu algoritmo as duas sugestões citadas no algoritmo LISTAORDEM (Item IV.6.1): Interromper as 'passagens' quando em uma delas não houver nenhuma inversão e em cada 'passagem' reduzir uma comparação.




Para cada turista de um verão em Florianópolis perguntou-se o número da praia de preferência (entre as 10 abaixo) e sua renda mensal.


Faça um algoritmo que imprima por praia, em ordem crescente de código, o número de turistas e a renda média destes turistas.



  Números das praias:

  1.- Jurerê        		5.- Lagoa da Conceição   		8.- Joaquina

  2.- Canasvieiras   		6.- Galheta              		9.- Armação

  3.- Santinho       		7.- Naufradados         		10.- Campeche

  4.- Ingleses


Fazer um algoritmo que:


a) Leia um conjunto X de n números, que representam alturas de pessoas, e calcule seu desvio padrão, DP:


	� EMBED Equation.2  ���			onde:  	� EMBED Equation.2  ���



b) Prova-se, em Estatística, que 67% dos elementos dos conjuntos chamados normais, estão um DP acima ou um DP abaixo da média X do conjunto. Conte as alturas das pessoas que estão no intervalo X + DP e verifique se de fato perfazem 67% das pessoas do conjunto.

c) Rearranje a coleção acima, de forma que o primeiro elemento          seja o menor.

d) Ordene o conjunto repetindo n-1 vezes, no máximo, o processo         programado em b).

e) Encontre as seguintes 'estatísticas' do vetor X:




Raiz Média Quadrática: RMQ = V (1/n)    Xi**2

Média Harmônica: MH = n/(1/x1+1/x2+1/x3+...+1/xn) = n/1/Xi

Média Ponderada: MP =    XiPi/   Pi

Amplitude Total: AT = máx(x1, x2, ...,xn) - mín(x1, x2, ...,xn)


(isto é, a diferença entre o maior e o menor elemento)



Média Geométrica: MG = V x1.x2. ... .xn

Mediana do vetor X: XM = X(n/2+1), se n é ímpar ou

	                       XM = (X(n/2)+X(n/2+1)), se n é par

(Isto é, mediana de X é a componente "do meio" de X. Ela existe sempre?)

Moda de X: É a componente de X que ocorre com maior frequência.




Leia 25 números em um registro formando o conjunto X e outros 25 no próximo registro, formando o conjunto Y. A seguir intercale os elementos de X e Y, formando um conjunto Z com 50 números. Deve ser impresso:


a) O conjunto Z.


b) Os elementos que estão na intersecção de X e Y.

c) A  maior diferença entre dois elementos consecutivos de X.

d) Encontre o grau de correlação linear (CR) entre os conjuntos X e Y, definida por:


� EMBED Equation.2  ���       (Atkinson, 523)

   Supondo X e Y como dois vetores relembre o cáclulo vetorial:

e) Calcule ¦X¦ = (x1**2 + x2**2 + ... + xn**2)**1/2


f) Calcule o produto escalar de X.Y = x1*y1 + x2*y2 + ... + xn*yn

g) Calcule o ângulo O entre X e Y, dado por:


        O = arc cos [X.Y/(¦a¦*¦b¦)]


a)
Uma forma de aproximar a integral (I = f(x)dx) é chamada "método trapezoidal", pois subdivide a área a integrar em n trapézios de mesma base, com comprimento h - consiste em dividir [a,b] em n subintervalos iguais e considerar:


	I =� EMBED Equation.2  ���h/2[f(x0) + 2f(x1) + 2f(x2) + ... + 2f(xn-1) + f(xn)]

onde h = comprimento de cada subintervalo e os xi são pontos internos oriundos da divisão (x0 = a  e xn = b: h = (b-a)/n). 

Elabore um algoritmo para calcular a integral I pelo método descrito anteriormente.

b) Outro método de integração numérica, conhecido como Regra de Simpson, é definido como:

� EMBED Equation.2  ���

onde f(x) é conhecida nos pontos xi, e xi = x0 + ih, para i= 0,1,2 ...,n, sendo n>=2 e par (logo, o valor da função deverá ser conhecido em um número ímpar de pontos xi, tais que h = xi - xi-1 seja constante).

Elabore um algoritmo que leia os valores xi e a seguir os valores f(xi). Suponha que os xi não estão ordenados, como exige a Regra de Simpson.

Use os algoritmos feitos em a) e b) para calcular numericamente as integrais I1 =  sen 2IIxdx,   I2 =  tg-1xdx,   I3 =  x e**xdx e I4=sen**2*xdx, no intervalo [a,b] (Obs.: se a=0 e b=II, então I4=II/2.




No item 4.6.1.- se fez o método de ordenação "da bolha" que, conforme cresce o número n de elementos a ordenar, cresce de um fator quadrático o número de comparações necessárias. Um método mais eficiente de ordenação, para um conjunto de n valores, seria o seguinte:


Conjunto�N(1)�N(2)�N(3)�..............�N(K)��Processo:

1º passo:

- "percorre-se" todo o conjunto, a partir de N(1), e encontra-se o menor elemento, suponha-se na posição i-ésima.

- Troca-se N(1) com o menor elemento, passando N(1) a ocupar a i-ésima posição.

- 2º passo:

- repete-se o passo 1, "percorrendo" o conjunto, agora a partir de N(2).

Assim, após n-1 passagens, o conjunto estará ordenado.

Programe o algoritmo proposto e após teste sua eficiência, em termos de tempo gasto, para um conjunto de, por exemplo, 1000 valores, e compare este tempo com o obtido com o método da "bolha".




Uma empresa é constituída de 5 filiais, numeradas de 1 a 5. Para cada empregado da empresa se tem um registro com o nome do empregado, número da filial e número do empregado. Sabendo-se que os registros não estão ordenados de nenhuma forma, fazer um algoritmo que imprima cinco tabelas, uma para cada cidade, por ordem crescente de número de cidade e, em cada cidade, por ordem crescente de empregado.





Dado um conjunto com N números inteiros maiores que 1 e menores que 25,


a) Construa um vetor de 25 posições, onde a i-ésima posição representa o número de vezes que o número i aparece no conjunto.

b) Leia dois inteiros, j e k, sendo k<=25, e insira j na k-ésima posição de N, deslocando os sucessores do k-ésimo elemento uma posição para frente.




Trinta alunos de uma turma fazem uma prova de 15 questões de múltipla escolha. Para cada aluno é feito um registro com seu nome, seguido de suas respostas. Antes dos 30 registros de resposta põe-se um cartão com o gabarito das 15 questões.


Fazer um algoritmo, usando vetores, que:

a) Se cada uma das 15 questões tem o mesmo valor (isto é, 10/15), imprima uma lista com os números de 1 a 30 e ao lado deles o nome de cada aluno com sua respectiva nota.

4.7.2
 
Matrizes



Para 200 dos municípios de Santa Catarina se tem a produção, em toneladas, de feijão, arroz, soja, milho, trigo, batata, pepino, banana, mandioca, sorgo, alfafa, aveia, cevada e abacaxi. Sabendo que para cada município foi criado um registro com seu nome e as produções dos itens acima, fazer um algoritmo que armazene os dados, em forma matricial, e a seguir imprima:


a) Produção global do município (soma de todos os produtos).


b) Nome do munícipio de maior produção global.

d) Nome do município que mais produziu pepino.

c) Ordenar a matriz por esta produção global (município de maior produção ocupa a 1a. linha, o segundo a linha 2, etc.)






Gerar matrizes A, B, C, D e E, de ordem n, tais que:


a) A: Tenha todos os elementos da diagonal principal iguais a 1 e os demais sejam nulos.


b) B: Tenha todos os elementos da diagonal secundária nulos e os restantes iguais a 1.

c) C: Seja uma matriz diagonal superior cujos elementos Cij valham Cij = 2i + 3j.


d)  D: Seja a soma de A e C.


e) E: Seja a própria matriz C, onde se permutou a linha 2 com a 4 e a coluna 1 com a 3.

f) Calcule a soma da diagonal principal de D, bem como a soma da diagonal secundária.

g) Calcule as somas parciais de todas as colunas de índices ímpares, considerando nestas colunas apenas as linhas de índices pares.

f) Transfira os elementos da diagonal principal da matriz D para o vetor DIAGPRIN.





Fazer um algoritmo que leia:


  - os valores m e n da matriz A(mxn), em um registro.

  - os valores n e m da matriz B(nxm), em um segundo registro.

  - os elementos de A e B, sendo uma linha em cada registro.

Deve ser impresso:

a) A matriz A', transposta de A.


a) O produto de A e B.

c) As linhas adjacentes r e s, de A, se existirem, que são proporcionais, isto é, se (Ar,j)/(As,j) = k (onde: s = r+1; k = constante; j = 1,n).






Numa eleição presidencial no Lisarb, país com 25 estados, 10 candidatos concorreram. O Tribunal Eleitoral de Lisarb contou os votos de cada candidato em cada estado. Faça um algoritmo que:


a) Leia uma matriz onde as linhas são os 25 estados e as colunas são os 10 candidatos à presidência. Os elementos desta matriz são os números de votos, nos 25 estados, dos 10 candidatos.



b) Totalize os votos de cada candidato e de cada estado, anexando mais uma linha e mais uma coluna à matriz.



c) Ordene a matriz de forma que a 1a. coluna seja a do candidato vencedor, a 2a. a do vice, até o "lanterna".






Trinta alunos de uma turma fazem uma prova de 15 questões de múltipla escolha. Para cada aluno é feito um registro com seu nome, seguido de suas respostas. Antes dos 30 cartões de resposta põe-se um registro com o gabarito das 15 questões.


Fazer um algoritmo, usando matrizes, que:



Se cada uma das 15 questões tem o mesmo valor (isto é, 10/15), 



imprima uma lista com os números de 1 a 30 e ao lado deles o nome de cada aluno, com sua respectiva nota.



Imprimir a mesma lista do item a), mas com as notas em ordem decrescente.


Alterar o algoritmo para que seja lido imediatamente após o registro gabarito, um segundo registro com os valores de cada uma das 15 questões (Note que agora o valor de cada questão é lido e é variável, diferentemente do Item a), onde era constante).


Comparar esta solução com a utilizada para vetores (Ex. 12, Item IV.7.1).





No exemplo discutido no Item IV.5.- supor que o fazendeiro deseja duas listagens ordenadas com:


a) Produções diárias em ordem decrescente, que deverá ficar na forma abaixo:



DIA�PRODUÇÃO��15�300��6�295��12�250��

b) Produções totais das vacas em ordem crescente, na forma:



VACA�PRODUÇÃO��3�150��7�180��1�182��

c) Imprimir o dia e a vaca que mais produziu.






Uma matriz A(n,n) é dita diagonal dominante se:


a)    ¦aij¦ <= ¦aii¦, i = 1,2,...,n



b)    ¦aij¦ <  ¦aii¦ para pelo menos um i.


Fazer um algoritmo que verifique se uma matriz An,n é diagonal dominante.


(Isto é, uma matriz é diagonal dominante se o somatório dos módulos dos elementos não diagonais é menor ou igual ao módulo do elemento diagonal; além disto, o dito somatório deve ser menor que o elemento diagonal em pelo menos uma das linhas: não poderá ser igual em todas).





Sabe-se qua a matriz B = A . A', onde A' é a transposta de A, é simétrica. Devido a erros de arredondamento, entretanto, raramente B é simétrica quando o cálculo é feito por computador. Dada a matriz A, verificar se B é simétrica, dentro de uma precisão relativa e dada.





Chama-se matriz de permutações a uma matriz quadrada unitária tal que em cada linha e em cada coluna exista apenas um elemento igual a 1.


a) Supondo que o elemento aij = 1, gere os restantes elementos de modo que uma matriz A seja de permutações.

b) Leia uma matriz M e verifique se ela é de permutações.




Faça um algoritmo para resolver um sistema linear AX = B, sabendo que a matriz A é triangular inferior, X e B são vetores com n elementos e A tem dimensão n, isto é, a equação acima tem a forma:





     
a11 ...................          x1          b1


     
a21 a22 ...............          x2          b2


     
a31 a32 a33 ...........          x3          b3


     
.


     
an1 an2 an3 an4 ...... ann       xn          bn





Modifique o algoritmo anterior para resolver um sistema linear   AX = B, onde A é triangular superior.





Um elemento de uma matriz que seja, simultaneamente, o mínimo de sua linha e máximo de sua coluna é chamado "ponto de sela" da matriz.


Faça um algoritmo que leia uma matriz A(m,n), encontre todos os seus pontos de sela, se existirem, e imprima-os, bem como sua localização (linha e coluna em que se encontram).





Faça um algoritmo que leia para cada vestibulando da UFSC deste ano, o nome e as notas obtidas em cada uma das provas, na ordem seguinte: Português, Matemática, Inglês, Física, Química, Geografia e História. Suponha que o número de vestibulandos é desconhecido, mas se sabe que é inferior a 20.000.


a) 
Calcule a média ponderada de cada vestibulando nas 7 provas, sabendo que cada prova tem peso definido pelo critério:


    
- Peso 3: para provas com      MG <  5


    
- Peso 4: para provas com 5 <= MG <  7


    
- Peso 5: para provas com 7 <= MG


Sendo MG: Média Global, média aritmética da respectiva matéria, entre todos os candidatos.


b) 
Imprima uma lista de nomes e médias, ordenada pelas médias ponderadas.





Colete os preços dos componentes e insumos do BAC (Bife a Cavalo), nos meses do janeiro, maio, setembro, em Fpolis.


Componentes: (1)150g de carne de primeira, (2)1 ovo médio, (3)1g de               pimenta, (4)3g de sal.

Insumo: (5)15g de gás de cozinha. Para considerar também os custos           da mão de obra, materiais de limpeza, óleo, etc. tome 50% do         preço carne, isto é, o preço de (6)75g como uma estimativa.

a) Formar uma matriz P(7x 13) em que as colunas 1, 5 e 9 sejam os preços lidos e as colunas restantes, até a 12a., são calculadas com base em uma inflação mensal média constante, estimada por você. Em cada linha se terá os preços de um item do BAC, nos 12 meses.

b) Na coluna 13 calcule a inflação do ano de cada item.

c) Na linha 7 totalize o custo de cada mês do BAC.

‘W’,170



‘W’,170



Pessoas com altura superior a média



‘X’,160



‘X’,160



xxxxxxxxx



xxxxxxxxxxx



‘A’,180



‘A’,180



DUPLICATAS










